algol,_i<

smex
30.9.69
Stellar Models
Homogeneous
Exercise Version, J. Otzen

_b_e_g_i_n
   _r_e_a_l pi, mod, recmo, G, k, m0, a, c, Ls, Rs, Ms, lTes, L, Te, R, lTe, lL, lR, L0, LV,
     eps, kappa, arb, arb1, U1, U3, cnsqm,epsp,epsn,Lneu, eps4,epsg,
     testg, testn, q, T, rho, P, T4, beta, age, M, X, Y, Z, qfit,
     XomH, YomHe, ZomA, eomua, eomue, tb, rb, sec,
     alter,
     delta, dd, ddr, dda, gr, alder, fsum, gsum, F, dpar, step, sres, mf,
     gacc, alfa, l, Hp, U, si, t1, ltb, lps, slT, slP,
     C1, C2, C3, C4,dummy,exp30;
   _i_n_t_e_g_e_r s, t, i, j, nm, t4, UNRW, UNWW, UNWP, UNWD, mode,
        g, autm, mtx4, ccore, num, rat, lat, ka,
        ztype, ilt, ilp, tsg, tsd;

   _a_r_r_a_y k1, y[1:5], XM[1:4], km[1:1,0:17,0:20], fmas[0:1],
      fp, lHJU, ddaa, lkap[0:30, 0:24], rltp, corr, res, ara[1:4], drdp[1:4, 1:4];

   _r_e_a_l _p_r_o_c_e_d_u_r_e antlog(r); _v_a_l_u_e r; _r_e_a_l r; antlog:= exp(r/mod);

   _r_e_a_l _p_r_o_c_e_d_u_r_e log(r); _v_a_l_u_e r; _r_e_a_l r; log:= mod×ln(r);

   _c_o_m_m_e_n_t _p_r_o_c_e_d_u_r_e LLGAUSS solves n linear equations with n unknowns;

   _p_r_o_c_e_d_u_r_e LLGAUSS(n,R,A,U,delta);
   _v_a_l_u_e n,delta;
   _i_n_t_e_g_e_r n;
   _r_e_a_l delta;
   _a_r_r_a_y R,A;
   _l_a_b_e_l U;
   _b_e_g_i_n
      _i_n_t_e_g_e_r i,j,k,h;
      _r_e_a_l a,b;
      k:= 0;

Do it again:
      k:= k+1;
      a:= A[k,k];
      h:= k;
      _f_o_r i:= k _s_t_e_p 1 _u_n_t_i_l n-1 _d_o
      _b_e_g_i_n
         b:= A[i+1,k];
         _i_f abs(a)>abs(b) _t_h_e_n _g_o_ _t_o S;
         h:= i+1;
         a:= b;
S:    _e_n_d;
      _i_f abs(a)<delta _t_h_e_n _g_o_ _t_o U;
      _i_f h=k _t_h_e_n _g_o_ _t_o eliminate;
      _f_o_r j:= k _s_t_e_p 1 _u_n_t_i_l n+1 _d_o
      _b_e_g_i_n
         b:= A[k,j];
         A[k,j]:= A[h,j];
         A[h,j]:= b
      _e_n_d;
eliminate:
      _f_o_r i:= k+1 _s_t_e_p 1 _u_n_t_i_l n _d_o
      _b_e_g_i_n
         b:= -A[i,k]/a;
         _f_o_r j:= k+1 _s_t_e_p 1 _u_n_t_i_l n+1 _d_o A[i,j]:= A[i,j]+A[k,j]×b
      _e_n_d;
      _i_f k<n _t_h_e_n _g_o_ _t_o Do it again;
next unknown:
      b:= 0;
      _f_o_r j:= n _s_t_e_p -1 _u_n_t_i_l k+1 _d_o
      b:=R[j]×A[k,j]+b;
      R[k]:= (A[k,n+1]-b)/A[k,k];
      k:= k-1;
      _i_f k>0 _t_h_e_n _g_o_ _t_o next unknown
   _e_n_d LLGAUSS;
   _p_r_o_c_e_d_u_r_e clock(print);
   _v_a_l_u_e print;
   _b_o_o_l_e_a_n print;
   _b_e_g_i_n
      _r_e_a_l c1,c2;
      _o_w_n _i_n_t_e_g_e_r last tt;
      _c_o_m_m_e_n_t
         1d 2h 3m 45.67s = 45.67 + 3×60 + 2×60×60 + 1×60×60×24 = 93825.67s;
      _p_r_o_c_e_d_u_r_e printit(c, text);
      _v_a_l_u_e c;

      _r_e_a_l c;
      _s_t_r_i_n_g text;
      _b_e_g_i_n
         _r_e_a_l d,h,m,s;
         _i_n_t_e_g_e_r id,ih,im;
         writecr;
         writetext(text);
         write(|<dddddddddd.dd|>,c);
         writetext(|<<s = |>);
         h:=c/3600;
         id:=entier(h/24);
         h:=h-id×24;
         ih:=entier(h);
         h:=h-ih;
         h:=h×60;
         im:=entier(h);
         h:=h-im;
         s:=h×60;
         _i_f id>0 _t_h_e_n
         _b_e_g_i_n
            writeinteger(|<p|>,id);
            writetext(|<<d |>)
         _e_n_d;
         _i_f ih>0 _t_h_e_n
         _b_e_g_i_n
            writeinteger(|<p|>,ih);
            writetext(|<<h |>)
         _e_n_d;
         _i_f im>0 _t_h_e_n
         _b_e_g_i_n
            writeinteger(|<p|>,im);
            writetext(|<<m |>)
         _e_n_d;
         write(_i_f s_>10 _t_h_e_n |<dd.dd|> _e_l_s_e |<d.dd|>,s);
         writetext(|<<s|>)
      _e_n_d printit;
      _c_o_d_e c1, c2;
      3, 45;
      3, 45;
      zl 0, grf pa1;
      zl 1, grf pa2;
      _e;
      _i_f print _t_h_e_n
      _b_e_g_i_n

         printit(c1, |<<GIER clock time: |>);
         printit(c2, |<<Wall-clock time: |>);
         writecr;
         writetext(|<<tracks transferred: |>);
         writeinteger(|<p|>,tracks transferred-last tt);
      _e_n_d print;
      last tt:=tracks transferred;
   _e_n_d clock;

   _c_o_m_m_e_n_t In the following block the tables are read to the arrays
        fp, HJU, ddaa, lkap, and km;

   _b_e_g_i_n
      _i_n_t_e_g_e_r k;
      _r_e_a_l t, p, t1;
      _p_r_o_c_e_d_u_r_e fpline(A,t);
      _a_r_r_a_y A;
      _s_t_r_i_n_g t;
      _b_e_g_i_n
         _i_f kbon _t_h_e_n
         _b_e_g_i_n
            writecr;
            writetext(|<<fpline |>);
            writetext(t);
         _e_n_d;
         _f_o_r j:= 0 _s_t_e_p 1 _u_n_t_i_l (_i_f k<ilp _t_h_e_n k _e_l_s_e ilp)  _d_o
         _b_e_g_i_n
            A[i,j]:= read real;
            _i_f kbon _t_h_e_n
            _b_e_g_i_n
               writecr;
               writetext(|<<A[|>);
               writeinteger(|<p|>,i);
               writetext(|<<,|>);
               writeinteger(|<p|>,j);
               writetext(|<<] := |>);
               write(|<-d.dddddd_{10}+d|>,A[i,j]);
            _e_n_d;
         _e_n_d;
         _f_o_r j:= ilp+1 _s_t_e_p 1 _u_n_t_i_l k _d_o
         _b_e_g_i_n
            dummy:= read real;
            _i_f kbon _t_h_e_n

            _b_e_g_i_n
               writecr;
               writetext(|<<dummy := |>);
               write(|<-d.dddddd_{10}+d|>,dummy);
            _e_n_d;
         _e_n_d;
      _e_n_d;
      UNRW:= 16; UNWW:= 17; UNWD:=9; UNWP:= 33;

NEWFP:
      dummy:=  select(UNWW);
      writetext(|<<
  t1, ilt:=  |>);
      t1:= read real;
      arb:= read real;
      dummy:= select(UNRW);
      ltb:=read real;
      ilt:=read integer;
      lps:=read real;
      ilp:=read integer;
      X:= read real;
      Y:= read real;
      Z:= read real;
      slT:= read real;
      dummy:=  read real;
      slP:= read real;
      dummy:=  read real;
label 1:
      t:= read real;
      p:= read real;
      _i_f t < t1-_{10}-4 _t_h_e_n
      _b_e_g_i_n
         ilp:= read integer;
         _f_o_r j:=1 _s_t_e_p 1 _u_n_t_i_l 4×(ilp+1) _d_o dummy:= read real;
         _g_o_ _t_o label 1
      _e_n_d;
      ltb:= t;
      lps:= p;
      ilt:=arb;
      ilp:=24;

      _f_o_r i:= 0 _s_t_e_p 1 _u_n_t_i_l ilt _d_o
      _b_e_g_i_n
         _i_f i|=0 _t_h_e_n

         _b_e_g_i_n
            t:= read real;
            p:= read real
         _e_n_d;

         k:= read integer;
         _i_f abs(t-ltb-slT×i)+abs(p-lps-slP×i)>_{10}-4 _t_h_e_n
         _b_e_g_i_n
            dummy:= select(UNWW);
            writetext(|<<
Error in fp-table|>);
            dummy:=  lyn;
            _g_o_ _t_o NEWFP
         _e_n_d;
         fpline(fp,|<<fp|>);
         fpline(lkap,|<<lkap|>);
         fpline(ddaa,|<<ddaa|>);
         fpline(lHJU,|<<lHJU|>);
      _e_n_d reading of fp tables;

ny tab:
      dummy:= select(UNWW);
      dummy:= lyn;
      dummy:= select(UNRW);
      tb:= read real;
      rb:= read real;
      ka:= read integer;
      t:= read integer;
      nm:= read integer;

      _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l nm _d_o
      _b_e_g_i_n
         XM[i]:= read real;
         _i_f kbon _t_h_e_n
         _b_e_g_i_n
            writecr;
            writetext(|<<XM[|>);
            writeinteger(|<p|>,i);
            writetext(|<<] := |>);
            write(|<-d.dddddd_{10}+d|>,XM[i]);
         _e_n_d;
         _f_o_r j := 0 _s_t_e_p 1 _u_n_t_i_l t _d_o
         _b_e_g_i_n
            _r_e_a_l read2;

            read2:=read real;
            _i_f kbon _t_h_e_n
            _b_e_g_i_n
               writecr;
               writetext(|<<read2|>);
               writetext(|<< := |>);
               write(|<  -d.dddddd_{10}+d|>,read2,abs(read2-tb-j×0.2),_{10}-4);
            _e_n_d;
            _i_f abs(read2-tb-j×0.2)>_{10}-4 _t_h_e_n
            _b_e_g_i_n
               writetext(|<<
TABELFEJL|>);
               _g_o_ _t_o ny tab
            _e_n_d;
            dummy:= read real;
            dummy:= read real;
            _f_o_r k:= 0 _s_t_e_p 1 _u_n_t_i_l ka-1 _d_o
            _b_e_g_i_n
               _i_f kbon _t_h_e_n
               _b_e_g_i_n
                  writecr;
                  writetext(|<<km[|>);
                  writeinteger(|<p|>,i);
                  writetext(|<<,|>);
                  writeinteger(|<p|>,j);
                  writetext(|<<,|>);
                  writeinteger(|<p|>,k);
                  writetext(|<<] := |>);
               _e_n_d;
               km[i,j,k]:= read real;
               _i_f kbon _t_h_e_n
               _b_e_g_i_n
                  write(|<  -d.dddddd_{10}+d|>,km[i,j,k]);
               _e_n_d;
            _e_n_d
         _e_n_d
      _e_n_d læs kappa;
   _e_n_d tablereading;

   dummy:= select(UNWW);
   writetext(|<<
STELLAR MODELS WITH OUTER ZONE
HOMOGENEOUS
EXERCISE VERSION
l/Hp:=  |>);
   alfa:= read real;

   _b_e_g_i_n
      _s_w_i_t_c_h SW:= choose, new model, new parameters, matrix, improove, frem;

      _c_o_m_m_e_n_t _p_r_o_c_e_d_u_r_e MERSN performs all integrations;
      _p_r_o_c_e_d_u_r_e MERSN;
      _b_e_g_i_n
         _r_e_a_l x0, ho3, eps, eq, gstep, tstop;
         _b_o_o_l_e_a_n last;
         _a_r_r_a_y y0, k3, k4, k5 [1:5];

         _p_r_o_c_e_d_u_r_e CRGR;
         _b_e_g_i_n
            _i_n_t_e_g_e_r t;
            _r_e_a_l gtg, teste, qe, qg, qn;

            _c_o_m_m_e_n_t _p_r_o_c_e_d_u_r_e CRGR determines where transitions between radiative and
              convective zones occur;

            t:= 0;
            gtg:= -testg;
            qe:= q;
            teste:= testn;
            qg:= x0;
            writechar(29);

iter:       _i_f abs(testn)>_{10}-4∧t<15 _t_h_e_n
            _b_e_g_i_n
               _i_f t>12 ∨ kb on  _t_h_e_n
               _b_e_g_i_n
                  _i_f ztype<3 _t_h_e_n WATM _e_l_s_e SKRIV;
               _e_n_d;
               qn:=(teste×qg-testg×qe)/(teste-testg);
               _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l nm _d_o y[i]:= y0[i];
               q:= x0;
               DIFF(k1);
               RKM(qn-x0);
               t:= t+1;
               _i_f t=14 _t_h_e_n writechar(62);
               DIFF(k4);

               _i_f sign(testn)=sign(teste) _t_h_e_n
               _b_e_g_i_n
                  qe:= qn;
                  teste:= testn
               _e_n_d
               _e_l_s_e
               _b_e_g_i_n
                  qg:= qn;
                  testg:= testn
               _e_n_d;
               _g_o_ _t_o iter
            _e_n_d;

            _i_f ztype<3 _t_h_e_n WATM _e_l_s_e SKRIV;
            writechar(62);
            DIFF(k1);
            _i_f abs(q- qfit)>2_{10}-7 _t_h_e_n last:= _f_a_l_s_e;
            testg:= testn:= abs(testn)×sign(gtg);
         _e_n_d procedure CRGR;

         _p_r_o_c_e_d_u_r_e RKM (h);
         _v_a_l_u_e h;
         _r_e_a_l h;
         _b_e_g_i_n
            ho3:= h/3;
            _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l nm _d_o y[i]:= k1[i]×ho3+y0[i];
            q:= x0+ho3;
            DIFF(k3);
            _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l nm _d_o y[i]:= (k1[i]+k3[i])/2×ho3+y0[i];
            DIFF(k3);
            _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l nm _d_o y[i]:= (k1[i]×.375+k3[i]×1.125)×ho3+y0[i];
            q:=x0+h/2;
            DIFF(k4);
            _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l nm _d_o y[i]:= (k1[i]×1.5-k3[i]×4.5+k4[i]×6)×ho3+y0[i];
            q:= h+x0;
            DIFF(k5);
            _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l nm _d_o y[i]:=((k1[i]+k5[i])/2+k4[i]×2)×ho3+y0[i];
         _e_n_d RKM;

         _p_r_o_c_e_d_u_r_e RKMER(h);
         _v_a_l_u_e h;
         _r_e_a_l h;

         _b_e_g_i_n
            x0:= q;
            _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l nm _d_o y0[i]:= y[i];
            RKM(h);
            eps:= 0;
            _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l (_i_f nm>4 _t_h_e_n 4 _e_l_s_e nm) _d_o
            _b_e_g_i_n
               eq:= abs(k1[i]×0.2-k3[i]×0.9+k4[i]×0.8-k5[i]×0.1);
               _i_f eq>eps _t_h_e_n eps:= eq
            _e_n_d;
            eps:= eps×abs(ho3);

            _i_f delta_>eps _t_h_e_n tsg:= tsg+1
            _e_l_s_e
            _b_e_g_i_n _c_o_m_m_e_n_t step unacceptable;
               _i_f kb on _t_h_e_n writetext(|<<.|>);
               tsd:= tsd+1;
               _f_o_r i:= 1 _s_t_e_p 1 _u_n_t_i_l nm _d_o y[i]:= y0[i];
               q:= x0;
               _g_o_ _t_o QR
            _e_n_d;
         _e_n_d RKMER;

         _c_o_m_m_e_n_t In the first part of the procedure body the outer zone is
           integrated;

         _i_f ztype<3 _t_h_e_n
         _b_e_g_i_n
            tstop:= fmas[1]/mod;
TOT6:       RKMER(step);
            DIFF(k1);
            _i_f sign(testn)= sign(testg)∧abs(testn)>_{10}-4 _t_h_e_n testg:= testn
            _e_l_s_e CRGR;
            _i_f kb on ∨ mode = 4 _t_h_e_n WATM;

            _i_f ztype=1 _t_h_e_n _g_o_ _t_o (_i_f y[1]<0.1 _t_h_e_n QR _e_l_s_e finis);
            _i_f ztype=2 _t_h_e_n _g_o_ _t_o (_i_f y[3]< tstop _t_h_e_n QR _e_l_s_e TILQ);

TILQ:       delta:= delta/_{10}-4;
IGEN:       _i_f abs(y[3] - y0[3])>_{10}-7 _t_h_e_n
            _b_e_g_i_n
               DIFF(k1);
               _i_f kb on _t_h_e_n WATM;
               RKMER((x0-q)×(y[3] - tstop)/(y[3] - y0[3]));

               _i_f abs(y[3] -tstop)>5_{10}-7 _t_h_e_n _g_o_ _t_o IGEN
            _e_n_d;
            delta:= delta×_{10}-4;
            _g_o_ _t_o finis;
         _e_n_d integration with ztype<3;

         _c_o_m_m_e_n_t In the second part of MERSNs body the inner zone is integrated
           (ztype = 3);

         nm:= _i_f mode=4 _t_h_e_n 5 _e_l_s_e 4;
         _g_o_ _t_o R;
try last:
         last:= _t_r_u_e;
         step:= qfit -q;
         _g_o_ _t_o R1;
R:       _i_f abs(step)>.05 _t_h_e_n step:= sign(step)×.05;
         _i_f abs( qfit-q)_<abs(step) _t_h_e_n _g_o_ _t_o try last;

         last:= _f_a_l_s_e;
R1:      RKMER(step);
         DIFF(k1);
         _i_f sign(testn)= sign(testg)∧ abs(testn)>_{10}-4 _t_h_e_n testg:= testn _e_l_s_e CRGR;
         _i_f kb on ∨ mode=4 _t_h_e_n SKRIV;
         _i_f last _t_h_e_n _g_o_ _t_o finis;

         _c_o_m_m_e_n_t In the last part of MERSN the new integration step is found;

         _i_f eps = 0 _t_h_e_n
         _b_e_g_i_n
            step:= 2×step;
            _g_o_ _t_o R
         _e_n_d;
QR:      step:= (delta/eps)|∧0.2×0.8×step;
         _g_o_ _t_o (_i_f ztype<3 _t_h_e_n TOT6 _e_l_s_e R);
finis:
      _e_n_d MERSN;

      _p_r_o_c_e_d_u_r_e DIFF(K);
      _a_r_r_a_y K;
      _b_e_g_i_n
         _i_n_t_e_g_e_r mt,mr,tad;

         _c_o_m_m_e_n_t In the first part of DIFF (_i_f ztype<3) the derivatives
           needed in the outer zone are computed;

         _i_f ztype<3 _t_h_e_n
         _b_e_g_i_n
            _i_n_t_e_g_e_r t,p;
            _r_e_a_l tm, pn, m, n, cmm1, cm0, cm1, cnm1, cn0, cn1, x2, FP, PmPr;

            _r_e_a_l _p_r_o_c_e_d_u_r_e INTP(A);
            _a_r_r_a_y A;
            INTP:= (A[t-1, p-1]×cnm1+A[t-1, p]×cn0+A[t-1, p+1]×cn1)×cmm1
                  +(A[t, p-1]×  cnm1+A[t, p]×  cn0+A[t, p+1]×  cn1)×cm0
                  +(A[t+1, p-1]×cnm1+A[t+1, p]×cn0+A[t+1,p+1]× cn1)×cm1;

            _p_r_o_c_e_d_u_r_e MIXING;
            _b_e_g_i_n
               _r_e_a_l U8,U16,fsi,fm,hs,s2;

               _c_o_m_m_e_n_t Here d ln T/ d ln P is computed according to the mixing - length
                    theory with l:= alfa × Hp;

               l:= alfa×Hp;
               U:= C3/eomua×Hp×T4/(T×exp(arb/mod)×FP×P/PmPr×kappa×rho×l×l);
               U8:= 8×U/9;
               U16:= 2×U8;
               hs:= U8×(ddr-dda);
la:            s2:= si×si;
               fsi:= s2×si+U8×s2+U16×U×si-hs;
               fm:= 3×s2+U16×(si+U);
               si:= si-fsi/fm;
               _i_f abs(fsi/hs)>1_{10}-6 _t_h_e_n _g_o_ _t_o la;
               dd:= dda+si×si+2×U×si;
            _e_n_d mixing procedure;

            P:= exp(q);
            _i_f ztype=1 _t_h_e_n
            _b_e_g_i_n
               T:=Te×(0.5+0.75×y[1])|∧0.25;
               y[3]:=ln(T)
            _e_n_d
            _e_l_s_e
            T:=exp(y[3]);
            tm:=(y[3]×mod-ltb)/slT;
            t:= entier(tm);
            pn:= (q×mod - (lps + tm×slP))/slP;
            p:= entier(pn);

            _i_f ilp > 2×p _t_h_e_n p:= p+1;
            m:= tm -t;
            n:= pn - p ;
            _i_f t < 1 ∨ t > ilt -1 ∨ p < 1 ∨  p > ilp - 1 _t_h_e_n
            _b_e_g_i_n
               writecr;
               write(|<-dddddd|>,t,ilt-1,p,ilp-1);
               CONTROL;
            _e_n_d;
            cmm1:= m/2×(m-1);
            cm0:= 1-m×m;
            cm1:= m/2×(m+1);
            cnm1:= n/2×(n-1);
            cn0:= 1-n×n;
            cn1:= n/2×(n+1);
            T4:=T×T;
            T4:= T4×T4;
            PmPr:= P-a/3×T4;
            FP:= INTP(fp);
            rho:= PmPr/FP/C2/T;
            kappa:= exp(INTP(lkap)/mod);
            dd:= ddr:= C1×kappa×P/(y[2]×T4);

            _i_f ztype=1 _t_h_e_n
            K[1]:= kappa×P/gacc
            _e_l_s_e
            _b_e_g_i_n
               x2:= y[1]×y[1];
               Hp:= x2×P/(y[2]×rho×gacc);
               dda:=INTP(ddaa);
               testn:=1/ddr -1/dda;
               _i_f ddr>dda _t_h_e_n
               _b_e_g_i_n
                  arb:= INTP(lHJU);
                  MIXING;
               _e_n_d;
               K[1]:= -Hp/(R×Rs);
               K[2]:= -C4×x2×rho×Hp;
               K[3]:= dd
            _e_n_d
         _e_n_d

         In the following second part of DIFF the derivatives needed in the inner
         zone are computed

         _e_l_s_e
         _b_e_g_i_n
            _r_e_a_l V,r3,T616,T626,T646,T6,t,r,f0,f1,n,p,p1,mnul,m1,Pc,ks;

            _c_o_m_m_e_n_t Now rho is determined through the equation of state;

            _c_o_d_e V;
            1, 45;
            zl 313
            _e;
            XomH:= X/1.00801;
            YomHe:= Y/4.0028;
            ZomA:= Z/16.942;
            eomue:= XomH + 2 × YomHe + Z/2;
            eomua:= XomH + YomHe + ZomA;
            T:= exp(y[3]);
            P:= exp(y[4]);
            V:= T|∧3;
            T4:= V × T;
            beta:= 1 - a/3 × T4/P;

            t:= P×beta/(T×k/m0);
            arb:= sqrt(V);
            rho:= t/(eomua+eomue×F);
            ks:= X+7×Z+3;
            r3:= cnsqm×ks|∧3;
DEGL:       Pc:= sqrt(r3×rho/T)×rho;
            t:= (P×beta+Pc)/(T×k/m0);
            f1:= rho;
            f0:= rho/(9.0476_{10}-9×arb)×eomue;
            r:= (f0 - 30)/30;
            F:= ((((0.109744×r-0.165376)×r+0.151420)×r-0.462698)×r+3.296236)×r+5.187420;
            rho:= t/(eomua + eomue×F);
            _i_f abs((f1-rho)/rho)>_{10}-5 _t_h_e_n _g_o_ _t_o DEGL ;

            V:= V/rho;
            dda:= 1/(4-1.5×beta×beta/(4-3×beta));
            t:= mod×y[3];

            _c_o_m_m_e_n_t In the following statement the energy generation rate is
                 determined;

            _i_f X<1_{10}-7 _t_h_e_n

            eps:= eps4:= 0
            _e_l_s_e
            _b_e_g_i_n
               T6:= T×_{10}-6;
               T616:= T6|∧(1/6);
               T626:= T616|∧2;
               T646:= T626|∧2;
               ks:= 0.1332_{10}9×sqrt(ks/V);
               arb:= 2.378_{10}-16×(1+0.00922×T626)×exp(-3×ks+99.96/T626 -30)×exp30×(X/(1-X-Z))|∧2;
               arb1:= sqrt(1+arb);
               arb:= (arb1+arb/2-1)/arb;
               epsp:= 4.199_{10}6×arb×(1+0.0123×T626+0.00781×T646+0.00067×T6)/T646×
                      exp(ks-33.809/T626)×rho×X|∧2;
               _i_f T6<12 _t_h_e_n
                  eps4:= 0
               _e_l_s_e
                  eps4:= 4.458_{10}27×(1+0.00274×T626)/T646×exp(7×ks-152.31/T626)×rho×X×Z;
               arb:= 1/(1+2.428_{10}16×X/(1+X)×(1+0.0041×T626)/T616×exp(4×ks-102.64/T626));
               eps:= epsp×(0.980+(0.964×arb-1.044)/(arb1+3))+eps4×0.936;
               epsn:=epsp+eps4-eps;
               _i_f T6<7 _t_h_e_n
                  eps:= _i_f T6>6.5 _t_h_e_n eps×(T6-6) _e_l_s_e eps/2;
            _e_n_d;

            epsg:= 0;
            eps:= eps+epsg;

            _c_o_m_m_e_n_t In the following the opacity is determined by interpolation in tables;

            r:= mod×ln(rho);
            arb:= (r-rb-3×(t-tb))×5;
            arb1:= (t-tb)×5;
            mt:= entier(arb1);
            mr:= entier(arb);
            _i_f mr<0 _t_h_e_n
            _b_e_g_i_n
               mr:= 0;
               KTABS
            _e_n_d;
            _i_f mr>ka-2 _t_h_e_n
            _b_e_g_i_n
               mr:= ka-2;
               KTABS
            _e_n_d;

            tad:= 0;
            p:= 1;
            f0:= km[tad+1,mt,mr]×p;
            f1:= km[tad+1,mt,mr+1]×p;
            n:= arb-mr;
            mnul:= f0+n×(f1-f0);
            f0:= km[tad+1,mt+1,mr]×p;
            f1:= km[tad+1,mt+1,mr+1]×p;
            m1:= f0+n×(f1-f0); n:= arb1-mt;
            kappa:= exp((mnul+n×(m1-mnul))/mod);

            _c_o_m_m_e_n_t Now the derivatives are easily written down;

            r:= exp(y[1]);
            r3:= r|∧3;
            K[1]:= U1×M/(r3×rho);
            K[2]:= Ms×M×eps/L0;
            K[5]:= Ms×M×epsn/L0;
            K[4]:= -G×(Ms×M/r|∧2)×q/P/(4×pi)×(Ms×M/r|∧2);
            ddr:= U3/M×kappa×y[2]×L0×P/(q×T4);
            testn:= 1/ddr-1/dda;
            K[3]:= K[4]×(_i_f ddr<dda _t_h_e_n ddr _e_l_s_e dda);
         _e_n_d diff for innerzone;
      _e_n_d procedure DIFF;

      _p_r_o_c_e_d_u_r_e KTABS;
      _b_e_g_i_n
         s:=select(UNWW);
         writetext(|<<
ext|>);
         dummy:= select(s);
      _e_n_d;

      _p_r_o_c_e_d_u_r_e CONTROL;
      _b_e_g_i_n
         dummy:= select(16);
         writetext(|<<
fp table too small  logT = |>);
         write(|<-d.dddd|>,y[3]×mod);
         writetext(|<<  logP = |>);
         write(|<-d.dddd|>, q×mod);
         writetext(|<<
Input new atm def|>);

         _g_o_ _t_o rep
      _e_n_d;

      _p_r_o_c_e_d_u_r_e WATM;
      _b_e_g_i_n
         writecr;
         _i_f ztype=2 _t_h_e_n
            write(|<-dd.dddd|>, y[2], y[1], 1, log(T), log(P), log(rho), log(kappa),
                   log(ddr), log(dda), log(dd))
         _e_l_s_e
            write(|<-dd.dddd|>, q×mod, log(T), log(rho), log(kappa), log(ddr), y[1]);
      _e_n_d WATM;

      _p_r_o_c_e_d_u_r_e ATMOF(lTe, lL);
      _v_a_l_u_e lTe, lL;
      _r_e_a_l lTe, lL;
      _b_e_g_i_n
         WTS;
         lR:= 2×(lTes - lTe) + lL/2;
         Te:= exp(lTe/mod);
         R:= exp(lR/mod);
         L:= exp(lL/mod);
         gacc:= G×M×Ms/(R×Rs)|∧2;y[2]:= 1;
         C1:= U3×Ls×L/M;
         C4:= 4×pi/gacc×G;
         eomua:= X/1.00801+Y/4.0028+Z/16.942;
         C2:= k×eomua/m0;
         _i_f kb on _t_h_e_n
         _b_e_g_i_n
            writecr;
            write(|<-dd.dddd|>, M, lTe, lL, lR, log(gacc), 5040/Te×2|∧0.25);
         _e_n_d;
         nm:= ztype:= 1;
         q:=(lps + (log(Te×0.5|∧0.25) - ltb)/slT×slP)/mod +0.01;
         y[1]:= 0;

ad1P:    DIFF(k1);
         _i_f kb on _t_h_e_n WATM;
         _i_f k1[1]<0.01 _t_h_e_n
         _b_e_g_i_n
            q:= q+1;
            _g_o_ _t_o ad1P
         _e_n_d;

         step:= 0.1;
         MERSN;
         _i_f kb on _t_h_e_n
         _b_e_g_i_n
            DIFF(k1);
            WATM
         _e_n_d;

         ztype:= 2;
         nm:= 3;
         y[3]:= ln(Te×(0.5+0.75×y[1])|∧0.25);
         y[1]:= 1;
         DIFF(k1);
         testg:= testn;
         _i_f kb on _t_h_e_n WATM;
         step:= 0.05;
         MERSN;

         _i_f kb on _t_h_e_n
         _b_e_g_i_n
            DIFF(k1);
            WATM
         _e_n_d;

         y[1]:= ln(y[1]×Rs×R);
         arb:= y[2];
         y[2]:= antlog(lL)×Ls/L0;
         y[4]:= q;
         F:= 1;
         q:= arb;
         ztype:= 3;
         Lneu:= y[5];
         y[5]:=0;
         DIFF(k1);
         _i_f kb on _t_h_e_n SKRIV;

         testg:= testn;
         step:= -0.0002;
         WTS
      _e_n_d ATMOF;

      _p_r_o_c_e_d_u_r_e WTS;
      _b_e_g_i_n
         _i_f kb on _t_h_e_n

         _b_e_g_i_n
            writecr;
            write(|<ddddd|>, tsg, tsd);
         _e_n_d
      _e_n_d WTS;

      _p_r_o_c_e_d_u_r_e CKUGL(Tc,rhoc);
      _v_a_l_u_e Tc,rhoc;
      _r_e_a_l Tc,rhoc;
      _b_e_g_i_n
         XomH:= X/1.00801;
         YomHe:= (1-X-Z)/4.0028;
         ZomA:= Z/16.942;
         eomue:= XomH + 2 × YomHe + Z/2;

         q:= fmas[0];
         y[1]:= 1;
         y[2]:= _{10}-4;
         ztype:= 3;
         y[3]:= Tc;
         T:= exp(Tc);
         rho:= exp(rhoc);
         arb:= F:= rho/(9.0476_{10}-9×exp(1.5×Tc))×eomue;
         F:= (F-30)/30;
         F:= ((((0.109744×F-0.165376)×F+0.151420)×F-0.462698)×F+3.296236)×F
               +5.187420;
         _i_f kb on _t_h_e_n
         _b_e_g_i_n
            write(|<-ddd.dddd|>,arb,F);
            writecr
         _e_n_d;
         y[4]:= ln(k/m0×rho×T×(XomH+YomHe+ZomA+eomue×F)+a/3×exp(Tc×4)
                 -sqrt(cnsqm×(X+7×Z+3)|∧3×rho/T)×rho);

         DIFF(k1);
         y[5]:= k1[5];
         _i_f kb on ∨  mode=4  _t_h_e_n SKRIV;
         step:= q:= fmas[0];
         y[1]:= ln(3×M×Ms×q/(4×pi×rho))/3;
         ddr:= U3×Ms×kappa×eps×P/T4;
         arb:= -G×M×Ms×q/2×rho/(exp(y[1])×P);
         y[2]:= eps×M×Ms×q/L0;
         y[3]:= Tc+arb×(_i_f ddr<dda _t_h_e_n ddr _e_l_s_e dda);
         y[4]:= y[4]+arb;

         DIFF(k1);
         y[5]:= (y[5]+k1[5])/2×q;
         testg:= testn;
         _i_f kb on ∨ mode=4  _t_h_e_n SKRIV;
      _e_n_d CKUGL;

      _p_r_o_c_e_d_u_r_e SKRIV;
      _b_e_g_i_n
         _i_f mode=4 _t_h_e_n _g_o_ _t_o udskriv;
         writecr;
         write(|<-dd.dddd|>,q,exp(y[1]-lR/mod)/Rs,y[2]×L0/(antlog(lL)×Ls),
               mod×y[3], mod×y[4], mod×ln(rho), mod×ln(kappa), log (ddr), eps);
         _g_o_ _t_o ud;
udskriv: writecr;
         write(|<-d.dddd|>, q, exp(y[1]-lR/mod)/Rs,y[2]×L0/(antlog(lL)×Ls),
               mod×y[3], 0.1×mod×y[4], mod×ln(rho), ddr, mod×ln(kappa),
               X, beta, eps4×0.936_{10}-4, eps×_{10}-4);
ud:   _e_n_d procedure SKRIV;

      _c_o_m_m_e_n_t In the first part of the program values are assigned to many variables;

      autm:= mode:= mtx4:= 0;
      gsum:= 0;
      mf:= 1;
      pi:= 3.14159265;
      mod:= 0.434294482;
      recmo:= 2.30258509;
      G:= 6.668_{10}-8;
      m0:= 1.66026_{10}-24;
      k:= 1.38046_{10}-16;
      a:= 7.5641_{10}-15;
      c:= 2.997929_{10}10;
      Ls:= 3.90_{10}33;
      Rs:= 6.9598_{10}10;
      Ms:= 1.989_{10}33;
      lTes:= log(Ls/(a×pi×c×Rs|∧2))/4;
      U1:= Ms/(4×pi);
      U3:= 3/(16×pi×a×c×G×Ms);
      cnsqm:= pi/k/8×(4.8029|∧3/m0/3_{10}30)|∧2/m0;
      exp30:= exp(30);
      fmas[0]:= 0.0003;
      fmas[1]:= 6;
      C3:= 6×sqrt(2)×a×c×m0/k;
      delta:= 0.0003;

      alter:= 0.005;

new model:
      writetext(|<<
M, X, Y, Z, qfit :=  |>);
      M:= read real;
      arb:= read real;
      arb1:= read real;
      LV:= read real;
      qfit:= read real;
      _i_f  arb|=X ∨ arb1|=Y ∨ LV|=Z _t_h_e_n
      writetext(|<<
X, Y, Z from the tables for the outer zone are different; outer zone values are
     used in the program|>);

new parameters:

      writetext(|<<
log Te, log L/Ls, log Tc, log rhoc :=  |>);
      lTe:= rltp[1]:= read real;
      lL:= rltp[2]:= read real;
      rltp[3]:= read real;
      rltp[4]:= read real;
      L0:= antlog(lL)×Ls;

rep:  writetext (|<<
M/Ms, X, Z = |>);
      write (|<-ddd.ddd|>, M, X, Z);
      writetext (|<<
log Te, log L/Ls, log Tc, log rhoc = |>);
      lR:= 2×(lTes-lTe) + lL/2;
      write(|<-dd.dddd|>, lTe, lL, rltp[3], rltp[4]);
      writecr;
      _i_f autm = 0 ∨ kb on _t_h_e_n _g_o_ _t_o SW[lyn];

      tsg:= tsd:= 0;

      clock(_f_a_l_s_e);
      CKUGL(rltp[3]/mod, rltp[4]/mod);
      MERSN;
      writecr;
      writetext(|<<CKUGL|>);
      writecr;
      clock(_t_r_u_e);

      _f_o_r i:= 1, 2, 3, 4 _d_o res[i]:= y[i];

      clock(_f_a_l_s_e);
      ATMOF(lTe, lL);
      MERSN;
      fsum:= 0;
      writecr;
      writetext(|<<ATMOF|>);
      writecr;
      clock(_t_r_u_e);
      writetext (|<<
res [i] = |>);
      _f_o_r i:= 1, 2, 3, 4 _d_o
      _b_e_g_i_n
         res[i]:= res[i] - y[i];
         write(|<-dd.dddd|>, res[i]);
         fsum:= fsum +abs(res[i])
      _e_n_d;

      writetext (|<<
fsum, tsg, tsd = |>);
      write(|<-dd.dddd|>,fsum);
      write(|<-dddd|>,tsg,tsd);
      writecr;
      writecr;

      _i_f autm = 0 _t_h_e_n
      _b_e_g_i_n
         i:= lyn;
         _i_f i = 51 _t_h_e_n
            _g_o_ _t_o matrix
         _e_l_s_e
            _g_o_ _t_o SW[i]
      _e_n_d
      _e_l_s_e
      _b_e_g_i_n
         mtx4:= mtx4-1;
         ccore:= -2;
         _i_f fsum<0.0005 _t_h_e_n _g_o_ _t_o frem;
         _i_f fsum<0.0050 _t_h_e_n
         _b_e_g_i_n
            ccore:= -3;
            _g_o_ _t_o improove
         _e_n_d;

         _i_f fsum>0.2×gsum∧mtx4<1 _t_h_e_n _g_o_ _t_o matrix
      _e_n_d;
      gsum:= fsum;

      _c_o_m_m_e_n_t In the following block 4 corrections to the parameters are found,
        and to each parameter is added the corresponding correction multiplied
        by a factor, mf;

improove:
      _b_e_g_i_n
         _a_r_r_a_y drdpa[1:4, 1:5];
         clock(_f_a_l_s_e);
         writecr;
         writetext(|<<start improove|>);
         writecr;
         _f_o_r s:= 1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o
         _b_e_g_i_n
            drdpa[s, 5]:= res[s];
            _f_o_r t:= 1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o drdpa[s, t]:= drdp[s, t]
         _e_n_d;
         LLGAUSS(4, corr, drdpa, exit, _{10}-6);
         _f_o_r s:= 1 _s_t_e_p 1 _u_n_t_i_l 4 _d_o rltp[s]:= rltp[s] - corr[s]×mf;
         lTe:= rltp[1];
         lL:= rltp[2];
         writecr;
         writetext(|<<improove|>);
         writecr;
         clock(_t_r_u_e);
         _g_o_ _t_o rep;
exit:    writetext(|<<
No solution|>);
         _g_o_ _t_o SW[lyn];
         _g_o_ _t_o rep;
      _e_n_d improove block;

      _c_o_m_m_e_n_t In the following block the coefficients in the equations for the
        corrections to the parameters are computed;
matrix:
      _b_e_g_i_n
         _i_n_t_e_g_e_r j;
         _a_r_r_a_y store res, store ys[1:4];

         _p_r_o_c_e_d_u_r_e ALTER(alterno);
         _i_n_t_e_g_e_r alterno;

         _c_o_m_m_e_n_t ALTER alters parameter no alterno, perform an testintegration, and
              computes the corresponding 4 coefficients in the equations for the
              parametercorrections;
         _b_e_g_i_n
            rltp[alterno]:= rltp[alterno] + alter;

OUTWARDS:   _i_f alterno<3 _t_h_e_n
            _b_e_g_i_n
               _f_o_r s:= 1, 2, 3, 4 _d_o res[s]:= store res[s] + store ys[s];
               _g_o_ _t_o INWARDS
            _e_n_d;
            CKUGL(rltp[3]/mod, rltp[4]/mod);
            MERSN;
            _f_o_r s:= 1, 2, 3, 4 _d_o res[s]:= y[s];

INWARDS:    _i_f alterno>2 _t_h_e_n
            _b_e_g_i_n
               _f_o_r s:= 1, 2, 3, 4 _d_o res[s]:= res[s] - store ys[s];
               _g_o_ _t_o ENDINTEGRATION
            _e_n_d;
            ATMOF(rltp[1], rltp[2]);
            MERSN;
            _f_o_r s:= 1, 2, 3, 4 _d_o res[s]:= res[s] - y[s];
ENDINTEGRATION:
            _f_o_r s:= 1, 2, 3, 4 _d_o drdp[s, alterno]:= (res[s] - store res[s])/alter;
            rltp[alterno]:= rltp[alterno] - alter;
         _e_n_d ALTER;

         _c_o_m_m_e_n_t The coefficients in the linear equations for the 4 parametercorrections
           are now computed by means of 4 testalterations of the parameters;

         writecr;
         writetext(|<<start matrix|>);
         writecr;
         clock(_f_a_l_s_e);
         _f_o_r j:= 1, 2, 3, 4 _d_o
         _b_e_g_i_n
            store res[j]:= res[j];
            store ys[j]:= y[j]
         _e_n_d;
         _f_o_r j:= 1, 2, 3, 4 _d_o ALTER(j);
         _f_o_r j:= 1, 2, 3, 4 _d_o res[j]:= store res[j];
         gsum:= 1;
         writecr;

         writetext(|<<matrix|>);
         writecr;
         clock(_t_r_u_e);
      _e_n_d matrix block;

      _i_f autm |= 0 _t_h_e_n _g_o_ _t_o improove;
      i:= lyn;
      _i_f i=57 _t_h_e_n _g_o_ _t_o improove _e_l_s_e _g_o_ _t_o SW[i];
      _g_o_ _t_o rep;
choose:  t:= select(UNWW);
      writetext(|<<
choose by letter:  |>);
      i:= lyn;
      writecr;
      writecr;
      writeinteger(|<p|>,i);
      _i_f i = 53 _t_h_e_n mode:= read integer; _c_o_m_m_e_n_t e;
      _i_f i = 54 _t_h_e_n
      _b_e_g_i_n
         fmas[0]:= read real;
         fmas[1]:= read real
      _e_n_d; _c_o_m_m_e_n_t f;
      _i_f i = 39 _t_h_e_n dpar:= read real; _c_o_m_m_e_n_t p;
      _i_f i = 18 _t_h_e_n _g_o_ _t_o stop; _c_o_m_m_e_n_t s;
      _i_f i = 36 _t_h_e_n mf:= read real;  _c_o_m_m_e_n_t m;
      _i_f i = 49 _t_h_e_n autm:= read integer; _c_o_m_m_e_n_t a;
      _i_f i = 52 _t_h_e_n delta:= read real; _c_o_m_m_e_n_t d;
      _i_f i = 57 _t_h_e_n _g_o_ _t_o improove; _c_o_m_m_e_n_t i;

      _i_f i = 51 _t_h_e_n
      _b_e_g_i_n _c_o_m_m_e_n_t c;
         _f_o_r s:= 1, 2, 3, 4 _d_o
         _b_e_g_i_n
            writecr;
            _f_o_r t:= 1, 2, 3, 4 _d_o write (|<-ddd.dddd|>, drdp[s,t])
         _e_n_d;
      _e_n_d;
      writecr;
      _g_o_ _t_o rep;

frem: writetext(|<<
log gacc, Te, Mbol = |>);
      write(|<ddd.dddd|>, log(gacc));
      write(|<ddddddd|>, antlog(lTe));

      write(|<-ddddd.ddd|>, 4.72 - 2.5×lL);
      _g_o_ _t_o choose;
stop:
      writecr;
      writetext(|<<tt: |>);
      writeinteger(|<p|>,tracks transferred);
   _e_n_d
_e_n_d;
t<